# This file is part of BlackArch Linux ( https://www.blackarch.org/ ).
# See COPYING for license details.

pkgname=python-winacl
_pkgname=winacl
pkgver=0.1.9
pkgrel=3
pkgdesc='Python library for interfacing windows security descriptors.'
arch=('any')
url='https://pypi.org/project/winacl/#files'
license=('MIT')
depends=('python' 'python-asn1crypto')
makedepends=('python-setuptools')
source=("https://files.pythonhosted.org/packages/source/${_pkgname::1}/$_pkgname/$_pkgname-$pkgver.tar.gz")
sha512sums=('8f668a61d82ad572f4fd65fc85578c465808cb19c33462ead9722ea7db5932341ba49b6ce021a185db0b6c729c32175ea29f1115bb1f7f3a3f119334a56fac9c')

build() {
  cd "$_pkgname-$pkgver"

  python setup.py build
}

package() {
  cd "$_pkgname-$pkgver"

  python setup.py install --root="$pkgdir" --prefix=/usr -O1 --skip-build
}
